Select into deyimi, Bloklar,Oracle akış  kontrol deyimleri


ORACLE (Devam) 
SELECT  …  INTO…  Deyim.

Sorgudan elde edilen değeri değişkene aktarmak.

SQL Server’da

SELECT @ortalama=avg(notu) from notlar

Yukarıda @ortalama değişkenine avg(notu) değeri atanır.

Oracle’da ise

SELECT avg(notu) INTO ortalama FROM notlar

Yukarıda ortalama değişkenine avg(notu) değeri atanır.

Örnek:

2053 nolu personelin adını soyadını değişkenlere aktarınız ve bu değişkenlerin değerini yazdırınız.

declare
    p_adi    varchar2(20);
    p_soyadi  varchar2(20);
begin
    select ad, soyad into p_adi, p_soyadi from personel where sicilno='2053';
    --yazdırmak dbms_output isimli paket var
    dbms_output.put_line(p_adi||' ' || p_soyadi); 
end;

PL/SQL Blokları
· Normal Blok

· Fonksiyon Bloğu

· Prosedür bloğu

Normal Blok Yapısı

Yukarıda yaptığımız uygulama bir normal blok uygulamasıdır.

Yapısı

declare 

      tanımlamalar

begin

      komutlar

end;

Örnek:
10 nolu departmanda çalışan personele %5 zam yapan kodu çalıştırınız. Zam oranı değişkeni tanımlayınız. Bu değişkene zam oranını atayınız.

Çözüm:
declare
   zam_orani number := 0.05;
begin
   update personel set maas=maas*(1+zam_orani) where departman=10;
   commit;
end;

Fonksiyon Bloğu (Oracle fonksiyonu)
Oracle üzerinde fonksiyon oluşturur.
create or replace function fonksiyonAdi (Parametreler)

return dönüşTipi

IS

Değişken tanımları

Begin

   Komutlar

   Return dönüşdeğeri

End;

Oracle fonksiyonların tablo dönmez. Oracle fonksiyonları scaler değerlidir.

Örnek:

Sicil numarası verilen personelin maaşını veren fonksiyon yazınız.

Çözüm:
create or replace function maas_bul(p_sicilno char)
-- parametrelerde veritipi uzunluğu yazılmaz
return number 
is
    p_maas number;
begin
    select maas into p_maas from personel where sicilno=p_sicilno;
    return p_maas;
end;
Şimdi bu fonksiyonu kullanalım. Yukarıdaki fonksiyondan 2053 nolu personelin maaşını yazdıralım.

declare
    maas number;
begin
    maas := maas_bul('2053');
    dbms_output.put_line(maas);
end;

prosedür bloğu
Oracle üzerinde prosedür oluşturur.

create or replace procedure prosedürAdi (parametreler)

IS

     Değişken tanımları

Begin

     Komutlar

End;

Örnek:

Personel tablosuna veri giren bir prosedür oluşturunuz.

create or replace procedure personel_veri_gir
(p_sicilno char, p_ad varchar2, p_soyad varchar2,p_maas number)
is
begin
     insert into personel(sicilno, ad, soyad, maas) values(p_sicilno, p_ad, p_soyad, p_maas);
     commit;
end;

Örnek veri girelim. Yukarıdaki prosedürü çalıştıralım.

begin
     personel_veri_gir(p_sicilno => '4444',p_ad => 'ali',p_soyad => 'veli',p_maas => 5000);
end;

ORACLE AKIŞ KONTROL DEYİMLERİ

IF…THEN…ELSE.. Deyimi

Genel ifadeler

IF koşul THEN

    Komutlar

END IF;

IF koşul THEN


Komutlar

ELSE


Komutlar

END IF;

IF koşul THEN


Komutlar

ELSIF koşul THEN


Komutlar

ELSE


Komutlar

END IF;

declare
  d_departman number;
  d_sicilno char(10):='2053';
begin
  select departman into d_departman from personel where sicilno=d_sicilno;
  if d_departman=5 then
     dbms_output.put_line('5 nolu departmanda');
  else
     dbms_output.put_line('diğer departmanlarda');
  end if;
end;

yukarıda if deyimi uygulaması yapıldı.

Yukarıdaki koda elsif ekleyelim.

declare
  d_departman number;
  d_sicilno char(10):='2053';
begin
  select departman into d_departman from personel where sicilno=d_sicilno;
  if d_departman=5 then
     dbms_output.put_line('5 nolu departmanda');
  elsif d_departman=10 then
     dbms_output.put_line('10 nolu departmanda');
  else
     dbms_output.put_line('diğer departmanlarda');
  end if;
end;

FOR DÖNGÜSÜ

Yapısı

FOR değişken IN deger1..deger2

LOOP


Komutlar

END LOOP;

Yukarıdaki kodda değişken değeri deger1den deger2’ye kadar birer artar.

Örnek:

For döngüsü ile 1 den 5’e kadar sayıları sırası ile yazdıralım.

declare

    i   number;

begin

    for i   in 1..5

    loop

       dbms_output.put(i||', ');

    end loop;

    dbms_output.put_line('');

end;
for döngüsünün geriye doğru sayması

FOR değişken IN REVERSE deger1..deger2

LOOP

      Komutlar

END LOOP;

Yukarı kodda deger2 den deger1 ‘e kadar geriye doğru sayılır.

Yukarıdaki örneği geriye saydıralım.

declare
    i   number;
begin
    for i in reverse 1..5
    loop
       dbms_output.put(i||', ');
    end loop;
    dbms_output.put_line('');
end;

if ve for döngüsü karışık bir örnek

declare

    i   number;

begin

    for i in reverse 1..5

    loop

       if i=1 then 

          dbms_output.put_line(i);

       else

          dbms_output.put(i || ', ');

       end if;   

    end loop;

end;
GOTO Deyimi

GOTO deyimi, program içinde akışın tanımlanan bir etikete(LABEL) gitmesini sağlar. Etiket <<etiketAdi>> şeklinde tanımlanır.

Kullanımı aşağıdaki gibidir.

begin

      …..

      <<etiketAdi>>

      ……

      GOTO etiketAdi

      …..

end;

bu deyim kullanılarak döngüden çıkılabilir.

Örnek:

For döngüsü 1 den 10’a kadar saysın, değişken değeri 5 olunca goto deyimi ile döngüden çıkılsın.

declare
    i  number;
begin
    for i in 1..10
    loop
        dbms_output.put_line(i);
        if i=5 then
            goto a;
        end if;
    end loop;
    dbms_output.put_line('Bu satır atlandı');
    
    <<a>>
    
    dbms_output.put_line('Dögüden çıkıldı');
end;

LOOP Döngüsü

Bu döngüden GOTO ve EXIT deyimleri ile çıkılır.

LOOP


Komutlar

END LOOP;

Örnek:

1 den 5’e kadar loop döngüsü ile sayıları yazdıralım.

Çözüm:

declare
   i   number(5);
begin
    i:=1;
    loop
        dbms_output.put_line(i);
        if i=5 then 
            exit;
        end if;
        i:=i+1;
    end loop;
end;  
WHILE Döngüsü

Koşul doğru olduğu sürece dönen bir döngüdür.

WHILE koşul

LOOP

      Komutlar

END LOOP;

declare
    i  number:=1;
begin
    while i<=5
    loop
        dbms_output.put_line(i);
        i:=i+1;
    end loop;
end;

EXIT deyimi
Döngüden çıkmayı sağlar.

EXIT       veya

EXIT WHEN koşul;

Koşul doğru olduğunda çıkış gerçekleşir.

EXIT WHEN i=5;
i değeri 5 olduğunda döngüde çıkar.

